Probelm 2: extended formulation for the asymmetric traveling salesman problem (ATSP)

1) **Strength of the ILP formulation with cut-set inequalities.**

Let $G = (V, E)$ be a complete directed graph, with a cost $c_{ij} \in \mathbb{R}$ for each arc $(i, j) \in E$. Consider the following Integer Linear Programming formulation (Dantzig, Fulkerson, Johnson, 1959) for the ATSP:

$$
\begin{aligned}
(DFJ) \qquad \min \quad & \sum_{(i,j)\in E} c_{ij}x_{ij} && && (1)\\
s.t. \quad & \sum_{j\in V: j\neq i} x_{ij} = 1 && i \in V && (2)\\
& \sum_{i\in V: i\neq j} x_{ij} = 1 && j \in V && (3)\\
& \sum_{(i,j)\in\delta^+(S)} x_{ij} \geq 1 && S \subset V, 1 \in S, |S| > 1 && (4)\\
& x_{ij} \in \{0, 1\} && (i, j) \in E, && (5)
\end{aligned}
$$

where the constraints (4) are the *cut-set inequalities*, and its linear relaxation:

$$
\begin{aligned}
(DFJ^0) \qquad \min \quad & \sum_{(i,j)\in E} c_{ij}x_{ij} && && (6)\\
s.t. \quad & \sum_{j\in V: j\neq i} x_{ij} = 1 && i \in V && (7)\\
& \sum_{i\in V: i\neq j} x_{ij} = 1 && j \in V && (8)\\
& \sum_{(i,j)\in\delta^+(S)} x_{ij} \geq 1 && S \subset V, 1 \in S, |S| > 1 && (9)\\
& x_{ij} \geq 0 && (i, j) \in E. && (10)
\end{aligned}
$$

a) Write with AMPL the original formulation (DFJ) and its linear relaxation (DFJ^0). A sketch of the file `.mod` is given on page 3 and following.

b) Compare the optimal solutions of (DFJ) and (DFJ^0). What do you observe?

2) **Compact extended ILP formulation.**

Consider the partial ILP formulation for the ATSP with only the assignment constraints:

$$\min \sum_{(i,j)\in E} c_{ij}x_{ij} \tag{11}$$

$$s.t. \sum_{j\in V:j\neq i} x_{ij} = 1 \quad i \in V \tag{12}$$

$$\sum_{i\in V:i\neq j} x_{ij} = 1 \quad j \in V \tag{13}$$

$$x_{ij} \in \{0,1\} \qquad (i,j) \in A. \tag{14}$$

c) Starting from the partial formulation (11)-(14), propose a compact extended formulation for ATSP. *Hint:* add, for each $i \in V$, an integer variable t_i representing the "position" in which node i is visited in the tour, and an appropriate set of constraints used to prevent all possible subtours.

d) Explain why such additional constraints exclude any subtour.

e) Is the integrality of variables t_i required?

f) The extended formulation of question c) is known in the literature as Miller, Tucker, Zemlin (MTZ) formulation. Write it in AMPL and find the solution of its linear relaxation for the given instance.

g) Compare the linear relaxation bound obtained with (MTZ^0) with that obtained with formulation (DFJ^0). What do you observe?

Solution sketch

The file `k6_dir.dat` contains the description of a complete graph with 6 nodes (optimal value of the objective function: 137):

```
#k6_dir.dat

data;

set V := a b c d e f;

param c:  a    b   c   d   e    f   :=
     a    .   03  05  100  88  55
     b   02    .  04   69  75  55
     c   04   03   .  100  99  85
     d   99   68  99    .  03  04
     e   87   74  98   02   .  05
     f   54   54  84   03  04   .
     ;
```

The file `dfj-i.mod` contains the partial model for formulation (DFJ), to be completed:

```
#dfj-i.mod

set V ordered;
param n := card {V};

set POWERSET := 0 .. (2**n - 1);

set S {k in POWERSET} := {i in V: (k div 2**(ord(i)-1)) mod 2 = 1};

set E := {i in V, j in V: ord(i) <> ord(j)};

#add parameters declarations

#add variables declarations

#add model
```

Observation: a generic subset $S \subseteq V$ can be represented as a vector of $n = |V|$ bits with value in $\{0, 1\}$, where position i has value 1 if the node i belongs to S, and 0 otherwise. Then the representation in base-2 of the integers between 0 and $2^n - 1$ gives all possible bit vectors of length n. Then, to build all possible subsets S of V, let

```
set POWERSET := 0 .. (2**n - 1);
```

be the set of indices from 0 to $2^n - 1$. We define the k-th subsets $S[k]$ with the binary representation of k: $S[k]$ contains the nodes whose corresponding bit is equal to 1. In base-2, the i-th figure of k_2 is equal to 1 if and only if

```
(k div 2**(ord(i)-1)) mod 2 = 1
```

then we can write all possible (exponentially many!) subsets as follows:

```
set S {k in POWERSET} := {i in V: (k div 2**(ord(i)-1)) mod 2 = 1};
```

The model can be solved running the file `dfj.run`:

```
model dfj.mod
data k6_dir.dat
option solver cplex;

solve;
display x;

printf "Linear relaxation\n\n";
option relax_integrality 1;

solve;
display x;
```

The file `MTZ-i.mod` contains the partial model of formulation (MTZ):

```
# SETS

set V ordered;

set V_reduced := {i in V : ord(i) > 1};

param n := card(V);

set E := {i in V, j in V: ord(i) <> ord(j)};

# PARAMETERS


# VARIABLES


# OBJECTIVE FUNCTION


# CONSTRAINTS
```

The model can be solved running the file `MTZ.run`:

```
model MTZ.mod
data k6_dir.dat
option solver cplex;

solve;
display x;
display t;

option relax_integrality 1;
solve;
display x;
display t;
```

Solution

a) The file `dfj.mod` contains the complete model:

```
# SETS

set V ordered;

param n := card(V);

set POWERSET := 0 .. (2**n -1);

set S{k in POWERSET} := {i in V : (k div 2**(ord(i)-1)) mod 2 = 1};

set E := {i in V, j in V: ord(i) <> ord(j)};


# PARAMETERS

param c{E}, >=0;


# VARIABLES

var x{E}, binary;


# OBJECTIVE FUNCTION

minimize cost:
        sum{(i,j) in E} c[i,j]*x[i,j];

# CONSTRAINTS

subject to degree_in{i in V}:
        sum{j in V: (j,i) in E} x[j,i] = 1;

subject to degree_out{i in V}:
        sum{j in V: (i,j) in E} x[i,j] = 1;

subject to cuts{k in POWERSET diff {2**n-1}: (k div 2**(1-1)) mod 2 = 1
                                and card(S[k]) > 1}:
        sum{i in S[k], j in V diff S[k]: (i,j) in E} x[i,j] >= 1;
```

And the solutions are as follow:

```
CPLEX 12.7.1.0: optimal integer solution; objective 137
9 MIP simplex iterations
0 branch-and-bound nodes
x [*,*]
:   a   b   c   d   e   f    :=
a   .   0   0   0   0   1
b   0   .   1   0   0   0
c   1   0   .   0   0   0
d   0   1   0   .   0   0
e   0   0   0   1   .   0
f   0   0   0   0   1   .
;

Linear relaxation

CPLEX 12.7.1.0: optimal solution; objective 137
24 dual simplex iterations (0 in phase I)
x [*,*]
:   a   b   c   d   e   f    :=
a   .   0   0   0   0   1
b   0   .   1   0   0   0
c   1   0   .   0   0   0
d   0   1   0   .   0   0
e   0   0   0   1   .   0
f   0   0   0   0   1   .
;
```

b) The formulation (DFJ) is so strong that for the considered small instance with six nodes, it is sufficient to solve its linear relaxation (DFJ^0) to obtain the integer optimal solution. It si worth empasizing that this is not true in general. However, the number of cut-set inequalities grows exponentially with the number of nodes, so that even for instances of few tens of nodes such a formulation is out of reach. In this case a cutting plane strategy, in which just a subset of cut-set inequalities are iteratively generated when needed, can be adopted (see the upcoming lectures).

c) Consider for each node i a variable t_i representig the order visit of i. The extended formulation is as follows

$$\min \sum_{(i,j)\in E} c_{ij}x_{ij} \tag{15}$$

$$\text{s.t. } \sum_{i\in V} x_{ij} = 1 \qquad j \in V \tag{16}$$

$$\sum_{j\in V} x_{ij} = 1 \qquad i \in V \tag{17}$$

$$t_1 = 1 \tag{18}$$

$$2 \le t_1 \le |V| \qquad i \in V, i \neq 1 \tag{19}$$

$$t_j \ge t_i + 1 - (|V| - 1)(1 - x_{ij}) \qquad (i,j) \in E, i \neq 1, j \neq 1 \tag{20}$$

$$x_{ij} \in \{0,1\} \qquad (i,j) \in E \tag{21}$$

$$t_i \text{ integer} \qquad i \in V, i \neq 1. \tag{22}$$

In this extended formulation we add a linear number of variables ($n = |V|$) and a polynomial number of constraints (in a complete graph, $(n-1)(n-2)$, in general $\mathcal{O}(n^2)$). Then it is a "compact formulation". Notice that constraints (18) and (19) are not necessary for the formulation to be correct, but just to ensure that variable t_i represents the order of visiting node i.

d) If a solution of (15)-(22) contained multiple subtours, then at least one of them would not contain node 1. If such a subtour, say C, contains l arcs, summing the constraints (20) associated to these arcs we would obtain

$$\sum_{i \in C} t_i \geq \sum_{i \in C} t_i + l,$$

which is a contradiction.

e) The integrality of variables t_i $\forall i \in V, i \neq 1$ is not necessary. Indeed the $|V| - 1$ variables t_i take value in the interval $[2, |V|]$, and since, for contraints (20), two consecutive values differ at least of a factor 1, variables t_i are forced to assume integer values. Clearly, however, alternative versions of this formulation are obtained by substituing constraints (20) by

$$t_j \geq t_i + \epsilon - (|V| - 1)(1 - x_{ij}) \quad (i, j) \in E, i \neq 1, j \neq 1,$$

for any $0 < \epsilon < 1$. In such cases, all subtours not containing node 1 are eliminated but the variables t_i can be fractional.

f) This is the AMPL model of the (MTZ) formulation

```
# SETS

set V ordered;

set V_reduced := {i in V : ord(i) > 1};

param n := card(V);

set E := {i in V, j in V: ord(i) <> ord(j)};

# PARAMETERS

param c{E}, >=0;


# VARIABLES

var x{E}, binary;

var t{V};

# OBJECTIVE FUNCTION

minimize cost:
        sum{(i,j) in E} c[i,j]*x[i,j];

# CONSTRAINTS

subject to degree_in{i in V}:
        sum{j in V: (i,j) in E} x[i,j] = 1;

subject to degree_out{i in V}:
        sum{j in V: (j,i) in E} x[j,i] = 1;

subject to order_visit{i in V_reduced, j in V_reduced:(i,j) in E}:
        t[i] - t[j] <= (n-1)*(1-x[i,j]) -1;
```

The solution of the linear relaxation is as follows:

```
CPLEX 12.7.1.0: optimal integer solution; objective 137
34 MIP simplex iterations
0 branch-and-bound nodes
x [*,*]
:   a   b   c   d   e   f    :=
a   .   0   0   0   0   1
b   0   .   1   0   0   0
c   1   0   .   0   0   0
d   0   1   0   .   0   0
e   0   0   0   1   .   0
f   0   0   0   0   1   .
;

t [*] :=
a  1
b  5
c  6
d  4
e  3
f  2
;

CPLEX 12.7.1.0: optimal solution; objective 20.2
22 dual simplex iterations (0 in phase I)

x [*,*]
:   a   b   c    d     e     f     :=
a   .   0   1   0     0     0
b   1   .   0   0     0     0
c   0   1   .   0     0     0
d   0   0   0    .    0.2   0.8
e   0   0   0   0.8    .    0.2
f   0   0   0   0.2   0.8    .
;

t [*] :=
a  1
b  3
c  2
d  2
e  2
f  2
;
```

Relaxing integrality, the optimal solution is no longer integer.

g) The linear relaxation (MTZ^0) appears to be easier to solve than (DFJ^0), since it is more compact (only polynomial number of constraints against exponantially many constraints). However, the results obtained for the six nodes instance, suggest that the lower bound on the minimum cost of the optimal tour, namely 20.2, is much weaker than the one provided by (DFJ^0), namely 137. This indicates that formulation (MTZ) is much weaker than (DFJ).